NAG Library Routine Document

G05SGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SGF generates a vector of pseudorandom numbers from an exponential mix distribution composed of m exponential distributions each having a mean a_i and weight w_i.

2 Specification

```
SUBROUTINE G05SGF(N, NMIX, A, WGT, STATE, X, IFAIL)
INTEGER           N, NMIX, STATE(lstate), IFAIL
double precision  A(NMIX), WGT(NMIX), X(N)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \sum_{i=1}^{m} \frac{1}{a_i} w_i e^{-x/a_i} \quad \text{if } x \geq 0,$$
$$f(x) = 0 \quad \text{otherwise,}$$

where $\sum_{i=1}^{m} w_i = 1$ and $a_i > 0$, $w_i \geq 0$.

G05SGF returns the values x_i by selecting, with probability w_j, random variates from an exponential distribution with parameter a_j.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SGF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: N ≥ 0.

2: NMIX – INTEGER *Input*

On entry: m, the number of exponential distributions in the mix.

Constraint: NMIX ≥ 1.

3: A(NMIX) – ***double precision*** array *Input*

On entry: the m parameters a_i for the m exponential distributions in the mix.

Constraint: A$(i) > 0.0$, for $i = 1, 2, \ldots,$ NMIX.

4: WGT(NMIX) – ***double precision*** array *Input*

On entry: the m weights w_i for the m exponential distributions in the mix.

Constraints:

$$\sum_{i=1}^{m} \text{WGT}(i) = 1.0;$$
$$\text{WGT}(i) \geq 0.0, \text{ for } i = 1, 2, \ldots, m.$$

5: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

6: X(N) – ***double precision*** array *Output*

On exit: the n pseudorandom numbers from the specified exponential mix distribution.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 0.

IFAIL $= 2$

On entry, NMIX ≤ 0.

IFAIL $= 3$

On entry, A(i) ≤ 0.0 for at least one A(i).

IFAIL $= 4$

On entry, WGT(i) < 0.0 for at least one WGT(i).

On entry, $\displaystyle\sum_{i=1}^{\text{NMIX}} \text{WGT}(i) \neq 1.0$.

IFAIL $= 5$

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints the first five pseudorandom numbers from an exponential mix distribution comprising three exponential distributions with parameters $a_1 = 1.0$, $a_2 = 5.0$ and $a_3 = 2.0$, and with respective weights 0.5, 0.3 and 0.2. The numbers are generated by a single call to G05SGF, after initialization by G05KFF.

9.1 Program Text

```
*     G05SGF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N, NMIX
      PARAMETER        (MSTATE=633,MSEED=1,N=5,NMIX=3)
*     .. Local Scalars ..
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION A(NMIX), WGT(NMIX), X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05SGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05SGF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameters A and WGT
      A(1) = 1.0D0
      A(2) = 5.0D0
      A(3) = 2.0D0
      WGT(1) = 0.5D0
      WGT(2) = 0.3D0
      WGT(3) = 0.2D0
*     Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*     Generate the variates
      IFAIL = 1
      CALL G05SGF(N,NMIX,A,WGT,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*     Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
   20 CONTINUE
```

```
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05SGF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05SGF Example Program Results

     0.4520
     2.2398
     1.4649
     0.2253
    11.2884
```